
#include p18f2550.inc 
 
 
 
; Various definitions 
 
 
; 3210 chip select pin 
#define tris__cs_3210        TRISB3 
#define _cs_3210             LATB3 
 
 
; 3210 reset and PCM bus pins 
 
#define tris_pcm_3210_pclk   TRISA,TRISA4   ; has to be set to 0 for output 
#define pcm_3210_pclk        LATA,LATA4     ; 1 is high, 0 is low 
 
#define tris_pcm_3210_drx    TRISA,TRISA3   ; has to be set to 1 for input 
#define pcm_3210_drx         PORTA,RA3     ; 1 is high, 0 is low 
 
#define tris_pcm_3210_dtx    TRISA,TRISA2   ; has to be set to 0 for output 
#define pcm_3210_dtx         LATA,LATA2     ; 1 is high, 0 is low 
 
#define tris_pcm_3210_fsync  TRISA,TRISA1   ; has to be set to 0 for output 
#define pcm_3210_fsync       LATA,LATA1     ; 1 is high, 0 is low 
 
#define tris__reset_3210     TRISA,TRISA0   ; has to be set to 0 for output 
#define _reset_3210          LATA,LATA0     ; has to be set to 0 for no-rst 
 
 
 
; Our exports 
 ; code 
 GLOBAL  tmr1_isr_init 
 GLOBAL  tmr1_isr 
 
 ; data 
 GLOBAL  rnginp 
 GLOBAL  rngout 
 GLOBAL  rinprp 
 GLOBAL  rinpwp 
 GLOBAL  routrp 
 GLOBAL  routwp 
 GLOBAL  elapsed 
 
 
 
; Globals, all placed in GPR1 (0x100). It is important to have all 
; of these in the same bank, because in the ISR we only do a single 
; BANKSEL, assuming that all data lie in the same bank, in order to 
; save us time. Thus, we ask the assembler to tell the linker to 
; place our data at the beginning of GPR1. 
gpr1  UDATA  0x100 
 
 ; next six globals implement two read/write rings for buffering PCM 
data 
 ; between the timer1 ISR and the main code; NOTE: DON't put anything 
 ; else before rnginp, otherwise the two rings will not be 64-byte 
 ; boundary aligned and nothing will work as expected 
rnginp RES  64  ; stores ~8 ms worth of incoming data 
rngout RES  64  ; stores ~8 ms worth of outgoing data 



rinprp RES  1  ; pointer to read from rnginp (main) 
rinpwp RES  1  ; pointer to write into rnginp (ISR) 
routrp RES  1  ; pointer to read from rngout (ISR) 
routwp RES  1  ; pointer to write into rngout (main) 
 ; next two values are one-byte buffers for the current bytes of PCM 
data 
 ; that are currently being read in and written out 
byteinp RES  1  ; stores the incoming 1-byte PCM data 
byteout RES  1  ; stores the outgoing 1-byte PCM data 
 ; next come two timers that are maintained by the ISR in order to do 
 ; different things at each invocation 
cnt8  RES  1  ; counts 8 isr invocations 
cnt4  RES  1  ; counts 4x8 invocations 
 ; and a debugging return value 
elapsed RES  1  ; stores @'s value (for debugging only!) 
 ; temporary storage for FSR1's value 
sfsr1hu RES  1  ; FSR1H, user's save space 
sfsr1lu RES  1  ; FSR1L, user's save space 
sfsr1hi RES  1  ; FSR1H, ISR's save space 
sfsr1li RES  1  ; FSR1L, ISR's save space 
 ; temporary storage for read/write 
rinprbu RES  1  ; byte being read from rnginp (user) 
routwbu RES  1  ; byte being written into rngout (user) 
 
 
 
; My TMR1 service interrupt routine 
 
; I have modified the bootloader code to jump to 0x820 for the 
; high-priority interrupt; thus, the PIC takes 3 instruction 
; cycles (I use C to denote an instruction cycle) to process 
; the interrupt and another 2 C to jump here, which sums up to 
; 5 C. 
 
timer1isr CODE  0x820   
 
tmr1_isr 
 
; DOCNOTE: 
; I use the @ symbol to denote the cycles already elapsed when we 
; enter this routine; this is equal to 5 (see previous note). 
 
 
 
       ; T:@ where @ >= 5, see above 
 
#ifdef STORE_ELAPSED 
 ; the following instruction saves @+1 into elapsed (this is a two- 
 ; word/two-cycle instruction which actually gets executed at the 
 ; second cycle, so an extra cycle is added to @ before we get its 
 ; value into elapsed). 
 ;;; MOVFF  TMR1L, elapsed  ; C:2 - comment me out! 
 ; NOTE: 
 ; use this ONLY for debugging! it adds two unnecessary clock cycles! 
 ; plus, these cycles are NOT taken into account in calculations for 
 ; the next value of the timer, so make sure to comment/if/ifdef this 
 ; out on 'production' versions, otherwise the clock will drift. 
#endif 
 



 
; Our first job is to raise PCLK 
 
 BSF  pcm_3210_pclk, ACCESS ; C:1 raise PCLK 
 
; Clear the interrupt flag (otherwise the call to the ISR loops 
forever!) 
 BCF  PIR1, TMR1IF  ; C:1 clear interrupt flag 
  
       ; T:@+2 
 
; Next job is to increment our counter modulo 8 and keep the status 
handy 
 
 BANKSEL  cnt8   ; C:1 use appropriate bank 
 ; Note: this is the only BANKSEL instruction used throughout the 
 ; code. All data are assumed to lie on the same bank. See note on 
 ; this at the UDATA section. 
 INCF  cnt8, F, BANKED  ; C:1 increment 8-step counter 
 MOVLW  7    ; C:1 modulo 8 
 ANDWF  cnt8, F, BANKED  ; C:1 save it 
 
       ; T:@+6  
 BNZ  notfirstof8   ; C:1/2  
 
; Our cnt8 has looped to 0; we need to increment our cnt4 and check 
; if we are starting a new 32-step sequence 
 
       ; T:@+7 
 
 INCF  cnt4, F, BANKED  ; C:1 
 MOVLW  3    ; C:1 
 ANDWF  cnt4, F, BANKED  ; C:1 
 
       ; T:@+10 
 
 BNZ  firstof4not32  ; C:1/2 
 
 
 ; following label is just for clarity, there's not any jump into it  
firstof32 
 
; At this point, we are starting a new 32-step sequence, thus we 
; need to raise FSYNC 
       ; T:@+11 
 
 BSF  pcm_3210_fsync, ACCESS ; C:1 
  
 
 
       ; T:@+12 
 
 ; during the cnt==0 round, we are doing PCM I/O, so start here 
 
 ; 3210's DRX signal is asserted on the falling edge of PCLK, 
 ; so we must prepare it before that 
 
 
 RLCF  byteout, F, BANKED ; C:1 rotate left through carry 
 BNC  fo32_clr_DRX  ; C:1/2 check if carry is set 
 BSF  pcm_3210_drx, ACCESS ; C:1 it is set, so set DRX to 1 



 BRA  fo32_DRX_ok   ; C:2 done 
 
fo32_clr_DRX     ; T:@+15 
 
 BCF  pcm_3210_drx, ACCESS ; C:1 carry not set, clear DRX 
 NOP      ; C:1 make both paths' delay eq. 
  
 
fo32_DRX_ok      ; T:@+17 via both paths 
 
 ; everything done, wait until @+18 (remember this cycle takes only 
23C) 
 
 NOP      ; C:1 
 
       ; T:@+18 
 
; At this point and assuming @==5, 23 cycles have elapsed since TMR1 
; fired, so it's time to lower PCLK and then FSYNC. TCy is 83.33+ns, 
; which is greater than 3210's 20ns minimum time t_h1 (cf. 3210 data 
; sheet, p.16), so we are on the safe side by lowering FSYNC right 
; after PCLK. 
 
; DOCNOTE: 
; similar to '@', which I use to denote the time when TMR1 fired 
; (hence PCLK^ is due) I use '&' to denote the time when PCLKv is 
; due, which is right now. 
 
       ; T: &+0 
 
; NOTE: I may need to add up to another 5 NOP (or other) instructions 
; here in order to adjust the duty cycle of PCLK closer to 50%, however 
; if they are not needed, I 'd rather avoid doing this, because these 
; cycles will be stolen from normal (non-ISR) execution. Another idea 
; would be to move the BCF instructions past the new timer setup, so 
; as to bring the duty cycle closer to 50%. 
 
 BCF  pcm_3210_pclk, ACCESS ; C:1 lower PCLK 
 BCF  pcm_3210_fsync, ACCESS ; C:1 lower FSYNC 
 
       ; T: &+2 
 
 SETF  TMR1H, ACCESS  ; C:1 set TMR1H to 0xFF 
 MOVLW  238    ; C:1 next interrupt at &+23 
 MOVWF  TMR1L, ACCESS  ; C:1 
  
       ; T: &+5 
 
 ; 3210's DTX signal is generated after the rising edge of PCLK 
 ; and maintained throughout the whole cycle until the next rising 
 ; edge; so it is ok collecting it here, after the falling edge, 
 ; and keeping it into the rightmost bit of byetin (it will get 
 ; shifted left on the next invocations) 
 
 CLRF  byteinp, BANKED  ; C:1 clear byteinp 
 
 BTFSC  pcm_3210_dtx, ACCESS ; C:1/2 skip next if DTX is 0 
 BSF  byteinp, 0, BANKED ; C:1 if 1, set bit 0 of byteinp 
  
       ; T:&+8 via both execution paths 
 



 RETFIE  FAST   ; C:2 
  
       
firstof4not32 
 ; this is the first cnt8 round of a cnt4 round other than the first 
one 
 
 ; we check further if cnt4==1 or 2; on cnt4==1, we transfer one byte 
 ; from the output ring into byteout, whereas on cnt4==2, we transfer 
 ; bytein into the input ring (we don't do anything special if cnt4==3) 
 
       ; T:@+12 
 
 DECF  cnt4, W, BANKED  ; C:1 decrement cnt4, rslt in W 
 BNZ  notsecondof4  ; C:1/2 if cnt4 > 1, jump on 
 
secondof4 
      ; T:@+14 
 ; this is the second cnt4 run just starting (cnt4==1); in this case, 
 ; we read one byte from the output ring and write it into byteout 
 
 ; save FSR1 to temp storage 
 MOVFF  FSR1H, sfsr1hi  ; C:2 save FSR1H 
 MOVFF  FSR1L, sfsr1li  ; C:2 save FSR1L 
 NOP      ; C:1 
 
 ; assuming @==5, 24 cycles have elapsed since TMR1 fireup; it's time 
 ; to lower PCLK and set the timer again, then we go on with reading 
 ; the output ring 
 
       ; T:@+19/& 
 
; See NOTE on additional NOPs that might be needed for adjusting PCLK duty 
cycle 
 
 BCF  pcm_3210_pclk, ACCESS ; C:1 lower PCLK 
 
       ; T:&+1 
 
 SETF  TMR1H, ACCESS  ; C:1 set TMR1H to 0xFF 
 MOVLW  237    ; C:1 next interrupt at &+23 
 MOVWF  TMR1L, ACCESS  ; C:1 
 
       ; T:&+4 
 
 ; fetch the read pointer, make sure it is not equal to the write ptr 
 MOVF  routrp, W, BANKED  ; C:1 
 SUBWF  routwp, W, BANKED  ; C:1 
 BZ  output_underrun  ; C:1/2 
 
 ; initialize FSR1 with the base address of the output ring 
 LFSR  FSR1, rngout  ; C:2 
 
       ; T:&+9 
 
 ; get the next byte from the ring into byteout, incr. pointer % 64 
 MOVF  routrp, W, BANKED  ; C:1 
 ADDWF  FSR1L, F, ACCESS  ; C:1 
 MOVFF  POSTINC1, byteout  ; C:2 
 MOVLW  0x3f    ; C:1 
 ANDWF  FSR1L, W, ACCESS  ; C:1 



 MOVWF  routrp, BANKED  ; C:1 
 
       ; T:&+16 
 
 ; restore FSR1 from temp storage 
 MOVFF  sfsr1hi, FSR1H  ; C:2 
 MOVFF  sfsr1li, FSR1L  ; C:2 
 
       ; T:&+20 
 
 ; just in time to return from our ISR at &+22, before TMR1 fires 
again! 
 
 RETFIE  FAST   ; C:2 
 
output_underrun 
 ; TODO: we may want to flag the underrun condition to the user space 
       ; T:&+10 
 
 RETFIE  FAST   ; C:2 
  
 
notsecondof4 
       ; T:@+15 
 
 DECF  W, W, ACCESS  ; C:1 
 BNZ  notthirdof4   ; C:1/2 
 
thirdof4 
 
 ; this is the third cnt4 run just starting (cnt4==2); in this case, 
 ; we write the contents of bytein into the input ring 
       ; T:@+17 
 
 ; check if the write pointer is equal to the read pointer minus one 
 ; (modulo 64 arithmetic) meaning that the ring is full; in such a 
 ; case we have an overrun condition 
 INCF  rinpwp, W, BANKED  ; C:1 increment the write ptr 
 ANDLW  0x3f    ; C:1 modulo 64 
       
       ; T:@+19/& 
 
 ; assuming @==5, 24 cycles have elapsed since TMR1 fireup 
 
 ; at this point we have to postpone our ring-writing work until 
 ; after lowering PCLK 
 
; See NOTE on additional NOPs that might be needed for adjusting PCLK duty 
cycle 
 
 BCF  pcm_3210_pclk, ACCESS ; C:1 lower PCLK 
 
       ; T:&+1 
 
 ; we take over our ring-writing work; W holds the incremented write 
ptr 
 
 SUBWF  rinprp, W, BANKED  ; C:1  
 BZ  input_overrun  ; C:1/2 
 
 



 
       ; T:&+3 
 ; save FSR1 to temp storage 
 MOVFF  FSR1H, sfsr1hi  ; C:2 save FSR1H 
 MOVFF  FSR1L, sfsr1li  ; C:2 save FSR1L 
 ; initialize FSR1 with the base address of the input ring, then add 
 ; the write pointer to it 
 LFSR  FSR1, rnginp  ; C:2 base rnginp address->FSR1 
 MOVF  rinpwp, W, BANKED  ; C:1 write pointer in W 
 ADDWF  FSR1L, F, ACCESS  ; C:1 add them, result in FSR1L 
 
       ; T:&+11 
 
 ; move byteinp (the PCM data collected) into the location pointed to 
 ; by FSR1, post-incrementing FSR1, then use the (incremented) FSR1L's 
 ; value modulo 64 as the new write pointer 
 MOVFF  byteinp, POSTINC1  ; C:1 move PCM data into ring 
 MOVLW  0x3f    ; C:1 prepare for modulo 64 
 ANDWF  FSR1L, W, ACCESS  ; C:1 incr. FSR1L mod 64 into W 
 MOVWF  rinpwp, BANKED  ; C:1 save as new rnginp wr ptr 
 
       ; T:&+15 
 
 ; restore FSR1 from temp storage 
 MOVFF  sfsr1hi, FSR1H  ; C:2 
 MOVFF  sfsr1li, FSR1L  ; C:2 
 
       ; T:&+19 
 
; At this point, we simply don't have enough time to set TMR1 (3 C) 
; and return (2 C) before &+23. So we just kill some time waiting, 
; then jump over to the ISR again. TMR1 should fire at &+23, which is 
; the same as our new @. Thus, now we are at @-4; we must wait 4 C 
; to reach @, then another 3 C to emulate PIC's 3C-delay before the 
; normal jump to 0x0008, then jump to the beginning of our ISR, which 
; takes another 2 C. 
  
 NOP      ; C:1 
 NOP      ; C:1 
 NOP      ; C:1 
 NOP      ; C:1 
 
       ; T:&+23=@ 
 
 NOP      ; C:1 
 NOP      ; C:1 
 NOP      ; C:1 
 
       ; T:@+3 
 GOTO  tmr1_isr   ; C:2 loop over, just at @+5 
 
 
 ; we arrive here if there was no space left in the input ring 
 
input_overrun 
       ; T:&+4 
 
 SETF  TMR1H, ACCESS  ; C:1 set TMR1H to 0xFF 
 MOVLW  240    ; C:1 next interrupt at &+23 
 MOVWF  TMR1L, ACCESS  ; C:1 
  



       ; T:&+7 
 
 ; TODO: we may want to flag the overrun condition to the user space 
 
 RETFIE  FAST   ; C:2 
  
 
 
notthirdof4 
 
       ; T:@+18 
 
 NOP      ; C:1 
 
 ; assuming @==5, 24 cycles have elapsed since TMR1 fireup 
 
       ; T:@+19/& 
 
; See NOTE on additional NOPs that might be needed for adjusting PCLK duty 
cycle 
 
 BCF  pcm_3210_pclk, ACCESS ; C:1 lower PCLK 
 
       ; T:&+1 
 
 SETF  TMR1H, ACCESS  ; C:1 set TMR1H to 0xFF 
 MOVLW  237    ; C:1 next interrupt at &+23 
 MOVWF  TMR1L, ACCESS  ; C:1 
 
       ; T:&+4 
 
 RETFIE  FAST   ; C:2 
  
 
notfirstof8 
       ; T:@+8 
 
 
 ; if this is the first cnt4 round (cnt4==0), we will do PCM data I/O 
 ; to and from byteout and byteinp, respectively  
 
 
 DECF  cnt4, W, BANKED  ; C:1 result->W, set carry if 0 
 BNC  nfo32_8nops   ; C:1/2 
 
       ; T:@+10 
 
 ; start by asserting the right DRX value (see previous notes on 
timing) 
 
 RLCF  byteout, F, BANKED ; C:1 rotate left through carry 
 BNC  nfo32_clr_DRX  ; C:1/2 check if carry is set 
 BSF  pcm_3210_drx, ACCESS ; C:1 it is set, so set DRX to 1 
 BRA  nfo32_DRX_ok  ; C:2 done 
 
nfo32_clr_DRX     ; T:@+15 
 
 BCF  pcm_3210_drx, ACCESS ; C:1 carry not set, clear DRX 
 NOP      ; C:1 make both paths' delay eq. 
 
nfo32_DRX_ok     ; T:@+15 via both paths 



 
 BRA  nfo32_2nops   ; C:2 
 
 
 ; we arrive here from branching if this is not the first cnt3 round 
       ; T:@+11 
nfo32_8nops 
 
 NOP      ; C:1 
 NOP      ; C:1 
 NOP      ; C:1 
 NOP      ; C:1 
 NOP      ; C:1 
 NOP      ; C:1 
 
       ; T:@+17 
nfo32_2nops 
 NOP      ; C:1 
 NOP      ; C:1 
 
 ; assuming @==5, 24 cycles have elapsed since TMR1 fireup 
 
       ; T:@+19/&+0 
       
; See NOTE on additional NOPs that might be needed for adjusting PCLK duty 
cycle 
 
 BCF  pcm_3210_pclk, ACCESS ; C:1 lower PCLK 
 
       ; T:&+1 
 
 SETF  TMR1H, ACCESS  ; C:1 set TMR1H to 0xFF 
 MOVLW  237    ; C:1 next interrupt at &+23 
 MOVWF  TMR1L, ACCESS  ; C:1 
 
       ; T:&+4 
 
 ; if this is our first cnt4 run, it's time to collect 3210's DTX 
signal 
 
 ; check if cnt4 is zero 
 DECF  cnt4, W, BANKED  ; C:1 result->W, sets carry if 0 
 BNC  nfo32_return  ; C:1/2 
 
 RLNCF  byteinp, F, BANKED ; C:1 rotate left byteinp 
 BTFSC  pcm_3210_dtx, ACCESS ; C:1/2 skip next if DTX is 0 
 BSF  byteinp, 0, BANKED ; C:1 if 1, set bit 0 of byteinp 
 
nfo32_return 
 
 ; done, return 
 RETFIE  FAST   ; C:2 
 
 
; Following code section is relocatable 
tmr1isrinit CODE 
 
; The TMR1 ISR initialization function 
tmr1_isr_init 
 
 ; disable TMR1 interrupts 



 BCF  PIE1, TMR1IE, ACCESS 
 
 ; (re-?) initialize all ISR data and outputs 
 BANKSEL  elapsed 
 CLRF  elapsed, BANKED  ; set elapsed to 0 
 MOVLW  7 
 MOVWF  cnt8, BANKED  ; set cnt8 to 7 
 MOVLW  3 
 MOVWF  cnt4, BANKED  ; set cnt4 to 3 
 
 BCF  pcm_3210_pclk, ACCESS ; make sure PCLK & FSYNC are 0 
 BCF  pcm_3210_fsync, ACCESS ; make sure PCLK & FSYNC are 0 
 
 ; reset read and write pointers for both rings to zero 
 CLRF  rinprp, BANKED 
 CLRF  rinpwp, BANKED 
 CLRF  routrp, BANKED 
 CLRF  routwp, BANKED 
 
 ; re-enable TMR1 interrupts 
 BSF  PIE1, TMR1IE, ACCESS 
 
 ; done 
 RETURN 
 
#ifdef UNIT_RING_WRITE 
 
; Reads one byte from the input PCM ring into W, returns with Z set if empty 
read_pcm_byte 
 
 BANKSEL  rinprp   ; should be the same 
throughout 
 
 ; check if the read pointer is equal to the write pointer (== ring is 
 ; empty) and return with the Z flag set if this is the case 
 MOVF  rinprp, W, BANKED 
 SUBWF  rinpwp, W, BANKED 
 BNZ  read_pcm_byte_notempty 
 RETURN 
 
read_pcm_byte_notempty 
 ; save FSR1 to temp storage 
 MOVFF  FSR1H, sfsr1hu 
 MOVFF  FSR1L, sfsr1lu 
 ; initialize FSR1 with the base address of the input ring 
 LFSR  FSR1, rnginp 
 ; fetch the read pointer 
 MOVF  rinprp, W, BANKED 
 ; add it to FSR1L 
 ADDWF  FSR1L, F, ACCESS 
 ; read the byte being pointed to into temporary storage, incr FSR1(L) 
 MOVFF  POSTINC1, rinprbu 
 ; update the pointer from incremented FSR1L modulo 64 
 MOVLW  0x3f   ; modulo 64 
 ANDWF  FSR1L, W, ACCESS 
 MOVWF  rinprp, BANKED  ; save back 
 
 ; make sure the Z flag is not accidentally set from the ANDWF above 
 IORWF  FSR1H, W, ACCESS ; FSR1H contains 1 for GPR1=>NZ 
 
 ; restore FSR1 from temp storage 



 MOVFF  sfsr1hu, FSR1H 
 MOVFF  sfsr1lu, FSR1L 
 
 ; reload the byte we just "read" from the input ring and place it 
 ; from temporary storage back into W 
 MOVF  rinprbu, W, BANKED 
 
 ; done 
 RETURN 
 
; Writes the byte currently in W into the output PCM ring, returns with Z 
set 
; if full 
write_pcm_byte 
 ; save W into temp storage 
 BANKSEL  routwp   ; should be the same 
throughout 
 MOVWF  routwbu 
 
 ; check if the write pointer is equal to the read pointer minus one 
 ; (modulo 64 arithmetic) meaning that the ring is full; in such a 
 ; case, return with the zero flag set 
 INCF  routwp, W, BANKED 
 ANDLW  0x3f 
 SUBWF  routrp, W, BANKED 
 BNZ  write_pcm_byte_notfull 
 RETURN 
 
write_pcm_byte_notfull 
 ; save FSR1 to temp storage 
 MOVFF  FSR1H, sfsr1hu 
 MOVFF  FSR1L, sfsr1lu 
 ; initialize FSR1 with the base address of the output ring 
 LFSR  FSR1, rngout 
 ; fetch the write pointer 
 MOVF  routwp, W, BANKED 
 ; add it to FSR1L 
 ADDWF  FSR1L, F, ACCESS 
 ; write the byt being pointed to from temporary storage, incr FSR1(L) 
 MOVFF  routwbu, POSTINC1 
 ; update the pointer from incremented FSR1L modulo 64 
 MOVLW  0x3f 
 ANDWF  FSR1L, W, ACCESS 
 MOVWF  routwp, BANKED  ; save it 
 
 ; make sure the Z flag is not accidentally set from the ANDWF above 
 IORWF  FSR1H, W, ACCESS ; FSR1H contains 1 for GPR1=>NZ 
  
 ; restore FSR1 from temp storage 
 MOVFF  sfsr1hu, FSR1H 
 MOVFF  sfsr1lu, FSR1L 
 
 ; done 
 RETURN 
#endif 
 
 
 END 


